Comparison Laplace and Z-transform

```
> restart:
```

Low pass filter equation

```
> EQ1 := Vin(t)  = R*i(t)+int(i(w),w=0..t)/C ;
  EQ2 := Vout(t) = int(i(w),w=0..t)/C ;
```

$$EQ1 := Vin(t) = R\,i(t) + \frac{\int_0^t i(w)\,dw}{C}$$

$$EQ2 := Vout(t) = \frac{\int_0^t i(w)\,dw}{C}$$

Load tranform library

```
> with(inttrans) ;
```

[*addtable, fourier, fouriercos, fouriersin, hankel, hilbert, invfourier, invhilbert, invlaplace, invmellin, laplace, mellin, savetable*]

Tranform the problem to the frequency domain (Laplace)

```
> EQ1s := laplace( EQ1, t, s ) ;
  EQ2s := laplace( EQ2, t, s ) ;
```

$$EQ1s := laplace(Vin(t), t, s) = R\,laplace(i(t), t, s) + \frac{laplace(i(t), t, s)}{C\,s}$$

$$EQ2s := laplace(Vout(t), t, s) = \frac{laplace(i(t), t, s)}{C\,s}$$

Syntactic simplification

```
> SUBS := { laplace(Vin(t), t, s)  = Vin(s),
            laplace(Vout(t), t, s) = Vout(s),
            laplace(i(t), t, s)    = i(s) };
```

$$SUBS := \{laplace(Vin(t), t, s) = Vin(s), laplace(Vout(t), t, s) = Vout(s), laplace(i(t), t, s) = i(s)\}$$

```
> EQ1s := subs(SUBS,EQ1s) ;
  EQ2s := subs(SUBS,EQ2s) ;
```

$$EQ1s := Vin(s) = R\,i(s) + \frac{i(s)}{C\,s}$$

$$EQ2s := Vout(s) = \frac{i(s)}{C\,s}$$

Isolate i(s)

```
> IS := solve( EQ1s, {i(s)} ) ;
```

$$IS := \left\{ i(s) = \frac{Vin(s)\,C\,s}{1 + R\,C\,s} \right\}$$

Evaluate transfer function

```
> EQTRASF := subs( IS, EQ2s ) ;
```

$$EQTRASF := Vout(s) = \frac{Vin(s)}{1 + R\,C\,s}$$

Sinusoidal input, evalute output

```
> EQOUT := subs( Vin(s)=laplace(cos(omega*t),t,s), EQTRASF ) ;
```

$$EQOUT := Vout(s) = \frac{s}{(s^2+\omega^2)\,(1+R\,C\,s)}$$

Simple fraction decomposition

```
> convert(EQOUT, parfrac, s) ;
```

$$Vout(s) = \frac{\omega^2 R\,C + s}{(1+\omega^2 R^2 C^2)\,(s^2+\omega^2)} - \frac{R\,C}{(1+\omega^2 R^2 C^2)\,(1+R\,C\,s)}$$

Non transitory part isolation

```
> non_transitory := op(1,op(2,%)) ;
```

$$non\_transitory := \frac{\omega^2 R\,C + s}{(1+\omega^2 R^2 C^2)\,(s^2+\omega^2)}$$

Invert laplace tranform of NON transitory part, the results is the asymptotic response to the signal cos (omega*t)

```
> SOLOUT := invlaplace( non_transitory, s, t ) ;
```

$$SOLOUT := \frac{\cos(\omega t) + \omega R\,C \sin(\omega t)}{1+\omega^2 R^2 C^2}$$

Rewrite in the form A*cos(omega*t+phi) ;

```
> TMP := collect( A*expand( cos(omega*t+phi) ) - numer(SOLOUT),
  {sin,cos} ) ;
```

$$TMP := (A\cos(\phi) - 1)\cos(\omega t) + (-A\sin(\phi) - \omega R\,C)\sin(\omega t)$$

```
> TMP1 := simplify(subs(sin(omega*t)=0,cos(omega*t)=1,TMP)) ;
  TMP2 := simplify(subs(cos(omega*t)=0,sin(omega*t)=1,TMP)) ;
```

$$TMP1 := A\cos(\phi) - 1$$

$$TMP2 := -A\sin(\phi) - \omega R\,C$$

Phase evaluation

```
> convert( subs( solve(TMP2,{A} ), TMP1 ), tan ) ;
```

$$-\frac{\omega R\,C}{\tan(\phi)} - 1$$

```
> SOLPHI := solve( convert( subs( solve(TMP2,{A} ), TMP1 ), tan ),
  {phi} ) ;
```

$$SOLPHI := \{\phi = -\arctan(\omega R\,C)\}$$

Modulus evaluation

```
> subs( solve( TMP2, {sin(phi)} ) union solve( TMP1, {cos(phi)} ),
        sin(phi)^2+cos(phi)^2 = 1) ;
  SOLA := solve( %, {A} )[1] ;
```

$$\frac{\omega^2 R^2 C^2}{A^2} + \frac{1}{A^2} = 1$$

$$SOLA := \left\{A = \sqrt{1+\omega^2 R^2 C^2}\right\}$$

Final solution as A*cos(omega*t+Phi)

```
> SOLFINAL := subs( SOLPHI union SOLA, A*cos(omega*t+phi)/denom
```

```
(SOLOUT) ) ;
```

$$SOLFINAL := \frac{\cos(\omega t - \arctan(\omega R C))}{\sqrt{1 + \omega^2 R^2 C^2}}$$

Some values for resistance and capacitance

```
> SUBSVAL := R = 10, C = 10^(-4), omega = 2*Pi*f ;
```

$$SUBSVAL := R = 10, C = \frac{1}{10000}, \omega = 2\pi f$$

```
> evalf(subs(SUBSVAL,2*Pi/(R*C))) ;
```

$$6283.185308$$

Plot transfer function

```
> with(plots):
```

Plot amplitute of the asymptotic response

```
> loglogplot( subs( SUBSVAL, 1/denom(SOLFINAL) ), f=0..10000) ;
```




```
> semilogplot( subs( subs( SUBSVAL, SOLPHI), phi),
             f=0..10000, numpoints=5000) ;
```



Transfer function evaluated on imaginary axes I*omega

```
> TRASF := evalc(subs(s=I*omega,subs( Vin(s)=1,rhs(EQTRASF)))) ;
```

$$TRASF := \frac{1}{1+\omega^2 R^2 C^2} - \frac{I\,\omega\,R\,C}{1+\omega^2 R^2 C^2}$$

Module of transfer function

```
> sqrt(simplify(evalc(Re(TRASF)^2+Im(TRASF)^2))) ;
```

$$\frac{1}{\sqrt{1+\omega^2 R^2 C^2}}$$

Phase of transfer function

```
> arctan(simplify(evalc(Im(TRASF)/Re(TRASF)))) ;
```

$$-\arctan(\omega\,R\,C)$$

Finite difference approzimation

Finite difference approximation of original ODE

```
> EQ1 := (Vin(k+1)-Vin(k))/h    = R*(i(k+1)-i(k))/h+i(k)/C ;
  EQ2 := (Vout(k+1)-Vout(k))/h = i(k)/C ;
```

$$EQ1 := \frac{Vin(k+1) - Vin(k)}{h} = \frac{R\,(i(k+1) - i(k))}{h} + \frac{i(k)}{C}$$

$$EQ2 := \frac{Vout(k+1) - Vout(k)}{h} = \frac{i(k)}{C}$$

Z tranform of recurrence

```
> EQ1z := ztrans( EQ1, k, z) ;
  EQ2z := ztrans( EQ2, k, z) ;
```

$$EQ1z := \frac{z\,ztrans(Vin(k), k, z) - Vin(0)\,z - ztrans(Vin(k), k, z)}{h}$$

$$= \frac{R\,(z\,ztrans(i(k), k, z) - i(0)\,z - ztrans(i(k), k, z))}{h} + \frac{ztrans(i(k), k, z)}{C}$$

$$EQ2z := \frac{z\,ztrans(Vout(k), k, z) - Vout(0)\,z - ztrans(Vout(k), k, z)}{h} = \frac{ztrans(i(k), k, z)}{C}$$

Syntactic simplification

```
> SUBS := { ztrans(Vin(k), k, z)  = Vin(z),
            ztrans(Vout(k), k, z) = Vout(z),
            ztrans(i(k), k, z)    = i(z) };
```

$$SUBS := \{ztrans(Vout(k), k, z) = Vout(z), ztrans(i(k), k, z) = i(z), ztrans(Vin(k), k, z) = Vin(z)\}$$

```
> EQ1z := subs(SUBS,EQ1z) ;
  EQ2z := subs(SUBS,EQ2z) ;
```

$$EQ1z := \frac{z\,Vin(z) - Vin(0)\,z - Vin(z)}{h} = \frac{R\,(z\,i(z) - i(0)\,z - i(z))}{h} + \frac{i(z)}{C}$$

$$EQ2z := \frac{z\,Vout(z) - Vout(0)\,z - Vout(z)}{h} = \frac{i(z)}{C}$$

Transfer function

```
> TRANS := subs(Vout(0)=0, Vin(0)=R*i(0), solve( subs( solve(EQ1z,
  {i(z)} ), EQ2z ), {Vout(z)} ))  ;
```

$$TRANS := \left\{ Vout(z) = \frac{h\,z\,Vin(z) - h\,Vin(z)}{z^2\,R\,C - 2\,R\,C\,z + z\,h + R\,C - h} \right\}$$

Sample time is h so that tk = k*h. Input singna a sample of cos(omega*t)

```
> SEGNALEIN := ztrans( cos(omega*h*k), k, z ) ;
```

$$SEGNALEIN := \frac{(z - \cos(\omega\,h))\,z}{z^2 - 2\,z\cos(\omega\,h) + 1}$$

Z-transform of signal

```
> SEGNALEOUT := simplify( subs( Vin(z)=SEGNALEIN, TRANS ) ) ;
```

$$SEGNALEOUT := \left\{ Vout(z) = \frac{h\,(z - \cos(\omega\,h))\,z}{(z^2 - 2\,z\cos(\omega\,h) + 1)\,(R\,C\,z - R\,C + h)} \right\}$$

```
> rhs(op(SEGNALEOUT));
```

$$\frac{h\,(z - \cos(\omega\,h))\,z}{(z^2 - 2\,z\cos(\omega\,h) + 1)\,(R\,C\,z - R\,C + h)}$$

```
> SOLTEMP := convert(subs(SEGNALEOUT,Vout(z)/z),parfrac,z) ;
```

$SOLTEMP :=$

$$\frac{h\,(-R\,C+h+\cos(\omega\,h)\,R\,C)\,R\,C}{(-2\,R^2\,C^2+2\,R\,C\,h-h^2+2\cos(\omega\,h)\,R^2\,C^2-2\cos(\omega\,h)\,R\,C\,h)\,(R\,C\,z-R\,C+h)}$$

$$-\left(\left(-R\,C\,z+C\,R\,z\cos(\omega\,h)+R\,C-2\,C\,R\cos(\omega\,h)^2+\cos(\omega\,h)\,R\,C+z\,h\right.\right.$$
$$\left.\left.-h\cos(\omega\,h)\right)h\right)\Big/\left(\left(-2\,R^2\,C^2+2\,R\,C\,h-h^2+2\cos(\omega\,h)\,R^2\,C^2\right.\right.$$
$$\left.\left.-2\cos(\omega\,h)\,R\,C\,h\right)\left(z^2-2\,z\cos(\omega\,h)+1\right)\right)$$

Find asymptotic solution (neglect temporary part)

```
> op(2,SOLTEMP);
```

$$-\frac{\left(-R\,C\,z+C\,R\,z\cos(\omega\,h)+R\,C-2\,C\,R\cos(\omega\,h)^2+\cos(\omega\,h)\,R\,C+z\,h-h\cos(\omega\,h)\right)h}{(-2\,R^2\,C^2+2\,R\,C\,h-h^2+2\cos(\omega\,h)\,R^2\,C^2-2\cos(\omega\,h)\,R\,C\,h)\,(z^2-2\,z\cos(\omega\,h)+1)}$$

```
> SOL := collect( invztrans(z*op(2,SOLTEMP),z,k), {sin,cos}) ;
```

$$SOL:=\frac{(-\cos(\omega\,h)\,R\,C\,h+R\,C\,h-h^2)\cos(\omega\,h\,k)}{(2\,R^2\,C^2-2\,R\,C\,h)\cos(\omega\,h)-2\,R^2\,C^2+2\,R\,C\,h-h^2}$$
$$-\frac{h\,R\,C\sin(\omega\,h)\sin(\omega\,h\,k)}{(2\,R^2\,C^2-2\,R\,C\,h)\cos(\omega\,h)-2\,R^2\,C^2+2\,R\,C\,h-h^2}$$

Find the coefficient of the conversion A*cos(x)+B*sin(x)=C*cos(x+phi)

```
> collect( A*cos(x)+B*sin(x)-expand(M*cos(x+phi)),{cos(x),sin(x)})
  ;
```

$$(A-M\cos(\phi))\cos(x)+(B+M\sin(\phi))\sin(x)$$

```
> solve( {A=M*cos(phi),B+M*sin(phi)}, {M,phi} ) assuming A::real,
  B::real ;
```

$$\left\{\phi=\arctan\left(-\frac{B}{\mathrm{RootOf}(-B^2-A^2+\_Z^2)},\frac{A}{\mathrm{RootOf}(-B^2-A^2+\_Z^2)}\right),M=\mathrm{RootOf}(-B^2-A^2\right.$$
$$\left.+\_Z^2)\right\}$$

```
> allvalues(%) assuming A::real, B::real ;
```

$$\left\{M=\sqrt{B^2+A^2},\phi=\arctan(-B,A)\right\},\left\{\phi=\arctan(B,-A),M=-\sqrt{B^2+A^2}\right\}$$

```
> AAA := (subs(cos(omega*h*k)=1,op(1,SOL)));
```

$$AAA:=\frac{-\cos(\omega\,h)\,R\,C\,h+R\,C\,h-h^2}{(2\,R^2\,C^2-2\,R\,C\,h)\cos(\omega\,h)-2\,R^2\,C^2+2\,R\,C\,h-h^2}$$

```
> BBB := (subs(sin(omega*h*k)=1,op(2,SOL)));
```

$$BBB:=-\frac{h\,R\,C\sin(\omega\,h)}{(2\,R^2\,C^2-2\,R\,C\,h)\cos(\omega\,h)-2\,R^2\,C^2+2\,R\,C\,h-h^2}$$

Find module of the transform

```
> WWW := simplify(sqrt(AAA^2+BBB^2)) assuming h > 0, R>0, C>0;
```

$$WWW:=h\sqrt{-\frac{1}{-2\,R^2\,C^2+2\,R\,C\,h-h^2+2\cos(\omega\,h)\,R^2\,C^2-2\cos(\omega\,h)\,R\,C\,h}}$$

```
> SUBSVAL := R = 10, C = 10^(-4), omega=2*Pi*f, h = 1/40000 ;
```

$$SUBSVAL:=R=10,\ C=\frac{1}{10000},\ \omega=2\,\pi\,f,\ h=\frac{1}{40000}$$

```
> loglogplot( subs(SUBSVAL, WWW), f=0..100000) ;
```




```
>
```